

Paper 243-2013

V is for Venn Diagrams

Kriss Harris, SAS Specialists Ltd

ABSTRACT

Would you like to produce Venn diagrams easily? This paper shows how you can produce stunning two-, three-, and four-way Venn diagrams by using the SAS® Graph Template Language, in particular the DRAWOVAL and DRAWTEXT statements. From my experience, Venn diagrams have typically been created in the pharmaceutical industry by using Microsoft Excel and PowerPoint. Excel is used to first count the numbers in each group, and PowerPoint is used to generate the two- or three-way Venn diagrams. The four-way Venn diagram is largely unheard of. When someone is brave enough to tackle it manually, then working out the numbers that should go in each of the 16 groups and inputting the right number into the right group is usually done nervously!

INTRODUCTION

Venn diagrams were introduced in 1883 by John Venn (1834-1923) the Hull born philosopher and mathematician. They are a great way to visualize elements that are unique to only one group and simultaneously visualize elements that intersect with other groups. They are symmetrical by nature and the number of groups in an order n Venn diagram = 2^n (including the group outside the diagram).

Venn Diagrams are built into JMP Genomics and they may be incorporated into the graphical procedures in SAS® 9.4, but until then, this paper will demonstrate a useful macro that can be used in SAS® 9.3.

MACRO CALL

The code below illustrates how the macro call works for the %VENN macro. The VENN_DIAGRAM macro variable, accepts the values 2, 3, or 4 and dependent on the value used produces the corresponding two-, three-, or 4 Way Venn diagram. In order for the macro to work, the macro variables GROUPA, and GROUPB must be given names, which will define the group names of the Venn diagram. In addition the output folder location and the output filename should be specified by using the macro variables OUT_LOCATION and OUTPUTFILENAME respectively. The commented code explains more detail on how to call the macro.

```
%macro venn(
data =
,venn_diagram = 2
/* Select whether you want a 2 Way, 3 Way or 4 Way Venn Diagram
EG for 2 way enter 2. Valid values are 2,3 and 4 */
,cutoff = < 0.3
/* Set the P Value cut-off or any other appropriate cut off */
,GroupA = Treatment A
/* Define group name 1, mandatory */
,GroupB = Treatment B
/* Define group name 2, mandatory */
,GroupC = Treatment C
/* Define group name 3, mandatory for 3 and 4 way diagrams */
,GroupD = Treatment D
/* Define group name 4, mandatory for 4 way diagrams */
,out_location = C:\SGF 2013\
/* Define the path for all output files e.g. C:\Venn Diagrams */
,outputfilename = Venn diagram
/* Define the filename for the graphic file */
);
```

LIMITATIONS

Presently the %VENN macro only works with quantitative data, and based on a cutoff specified, in the CUTOFF macro variable determines how the groups in the Venn Diagram overlap.

METHOD

Essentially the macro works by counting the number in each group, and then once this is done SAS® Graph Template language is used to produce the Venn Diagrams with the corresponding numbers in each group displayed. I am more excited about how Graph Template Language is used to produce the Venn Diagram, so I will concentrate on the mechanics of this in the paper.

If you have already calculated or got an idea of the numbers that you would like to display on the Venn Diagram, then you will only need to extract the code at the later stages of the macro, that is where the Graph Template Language and SGRENDER procedure is used, and substitute your numbers into the macro variables A, B, AB, TO, etc that appear later in the program.

GRAPH TEMPLATE LANGUAGE

BLANK CANVAS

Firstly a data set called test was created, that had two variables x and y and had values ranging from 1 to 100. This was created so that it could be used as a blank canvas in conjunction with the SCATTERPLOT statement. I will explain this more later.

```
data test;
  do x = 1 to 100;
    y = x;
    output;
  end;
run;
```

TWO WAY VENN DIAGRAM

GRAPH TEMPLATE LANGUAGE

To produce the Venn Diagram the LAYOUT OVERLAY container was used and contains the SCATTERPLOT plot statement, and DRAWOVAL and DRAWTEXT draw statements. The yaxis and xaxis label, line, tick marks and tick values are of no interest when producing the Venn Diagram, so these are suppressed in the LAYOUT OVERLAY statement by using the option DISPLAY = NONE.

The SCATTERPLOT statement is used to essentially produce the blank canvas for our Venn Diagram masterpiece by hiding the markers to with the SIZE = 0 marker attributes options.

The fun part begins with the DRAWOVAL draw statements, which basically draw the circles. The X and Y arguments are where the center of the oval is going to be drawn. As you recall the scale we used for the blank canvas goes from 0 to 100 for the xaxis and yaxis, and the X and Y arguments that we apply map directly to the values on our scale, so in our first example the centre point will be precisely at X = 36 and Y = 50. In our example the WIDTH = 45 and the HEIGHT = 60 for the oval. These are specified like that because it means that the ratio between the WIDTH and HEIGHT of the oval is 3:4 and this is to compensate for the fact that usually the default graph and also in our case the ratio between the WIDTH and HEIGHT of the entire graph is 4:3. Therefore the ratio that we use for the oval compensates for this and enables circles to be produced instead of ovals.

The DRAWTEXT draw statement does exactly what it says and it displays the numbers in the Venn Diagram groups and displays the group names.

Please see below for the example Graph Template Language code.

```

proc template;
  define statgraph Venn2Way;
    beginngraph / drawspace=datavalue;

      /* Plot */
      layout overlay / yaxisopts = (display = NONE) xaxisopts = (display
      = NONE);
      scatterplot x=x y=y / markerattrs=(size = 0);

      /* Venn Diagram (Circles) */
      drawoval x=36 y=50 width=45 height=60 /
      display=all fillattrs=(color=red)
      transparency=0.75 WIDTHUNIT= Percent HEIGHTUNIT= Percent;

      drawoval x=63 y=50 width=45 height=60 /
      display=all fillattrs=(color=green)
      transparency=0.75 WIDTHUNIT= Percent HEIGHTUNIT= Percent;

      /* Numbers */
      drawtext "&A" / x=33 y=50 anchor=center;
      drawtext "&AB" / x=50 y=50 anchor=center;
      drawtext "&B" / x=66 y=50 anchor=center;
      drawtext "Outside Union - &TO" / x=50 y=10 anchor=center width
      = 30;


      /* Labels */
      drawtext "&GroupA" / x=30 y=15 anchor=center width = 30;
      drawtext "&GroupB" / x=70 y=15 anchor=center width = 30;

    endlayout;
  endngraph;
end;
run;

```

Figure 1 Two Way Venn Diagram

This is an example of how the Two Way Venn Diagram looks.



THREE- AND FOUR WAY VENN DIAGRAMS

To produce three way Venn Diagrams we just need to add on what we had for the two way Venn Diagram Graph Template Language code. Essentially this is an extra DRAWOVAL statement for the third circle, and five more

DRAWTEXT statements for the additional numbers in each group and label accompanied with different X and Y locations. As you can envisage the four way Venn Diagram needs an extra DRAWOVAL statement compared to the three way, and a lot more DRAWTEXT statements. To produce the ellipses we can use a 1:4 ratio between the WIDTH and the HEIGHT in the DRAWOVAL statement, and the ROTATE argument can be used to position the ellipses in the right direction.

In the end trial and error was used to determine the location, size and rotation of the four way Venn Diagrams, but I believe there has to be a mathematical way to obtain the exact locations and dimensions for the optimal four way Venn Diagram.

Figure 2 Three Way Venn Diagram





Figure 3 Four Way Venn Diagram



CONCLUSION

Venn Diagrams are very useful for visualizing the relationships between groups. This poster has demonstrated how the technique can be implemented using SAS 9.3, with aesthetically pleasing results.

This poster is an updated version of a previous Venn Diagram paper I presented in 2008, and the improvements are in the quality of the Venn Diagram image and the consistency of how the image is generated. This is because the previous macro was developed for SAS 9.1.3 and a template was not used as in this one to allow for a consistent output of the generation of the Venn Diagrams.

Further upgrades can be made to allow the macro to accept qualitative data, to drill down, and to generate proportional Venn Diagrams. Also to reduce the font size if the numbers in the groups are very large. In my opinion the most pertinent update will be to allow the macro to accept qualitative data and hence work out the exact numbers in each group from a qualitative data set. I did explain a way around this, and I know of some websites where one can obtain numbers for a two- and three way Venn Diagram. I have not come across this yet for a four way Venn Diagram and I know from experience that calculating the elements in each group manually for a four way Venn Diagram is not easy. Therefore in my next update of the macro, I will incorporate the use of qualitative data.

REFERENCES

- Harris, Kriss. 2008. "How To Generate 2, 3 and 4 Way Venn Diagrams with Drill Down Functionality within 4 minutes!" Proceedings of the SAS Global Forum 2008 Conference. Available at <http://www2.sas.com/proceedings/forum2008/073-2008.pdf>
- "SAS® 9.3 Graph Template Language Reference". Available at <http://support.sas.com/documentation/cdl/en/grstatgraph/65377/HTML/default/viewer.htm#titlepage.htm>
- "Venn Diagram". Available at http://en.wikipedia.org/wiki/Venn_diagram

ACKNOWLEDGMENTS

I would like to thank Sharon Carroll for the support she gave me whilst preparing the paper and I would like to thank Adam Taylor for prompting me to start the Venn diagram SAS Macro years ago by simply asking me if it was possible to do Venn diagram calculations in SAS.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Kriss Harris
 Enterprise: SAS Specialists Ltd
 E-mail: italjet125@yahoo.com
 Web: <http://www.krisssharris.co.uk>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.

APPENDIX

```

/* Venn Diagram Macro */
data data;
call streaminit(123);
do i = 1 to 1000;
id = i;
A = RAND('UNIFORM');
B = RAND('UNIFORM');
C = RAND('UNIFORM');
D = RAND('UNIFORM');
output;
end;
run;

%macro venn( data =
,venn_diagram = 2          /* Select whether you want a 2 Way, 3 Way or 4 Way Venn Diagram. EG for 2
way enter 2. Valid values are 2,3 and 4 */

```

```

,cutoff = < 0.3          /* Set the P Value cut-off or any other appropriate cut off
                          Valid values are the right hand side of an if statement */

,GroupA = Treatment A /* Define group name 1, mandatory */
,GroupB = Treatment B /* Define group name 2, mandatory */
,GroupC = Treatment C /* Define group name 3, mandatory for 3 and 4 way diagrams */
,GroupD = Treatment D /* Define group name 4, mandatory for 4 way diagrams */

,out_location = C:\Documents and Settings\yex7977\My Documents\SGF 2013\
                /* Define the path for all output files e.g. C:\Venn Diagrams */

,outputfilename = Venn diagram Test /* Define the filename for the graphic file */
,drilldownfilename = Drilldown      /* Define the filename for the drill down data file */

);

/* Calculate the category for each observation in the dataset
   This has to be done differently for 2,3 and 4 way diagrams */

data data_reformatted;
  set &data;
run;

/* Counting the overlap */

data data_reformatted2;
set data_reformatted;

%IF &venn_diagram = 2 %THEN %DO;

  if A ne . and B ne . then do;
    if A &cutoff and B &cutoff then AB = 1;
    else AB = 0;
  end;

  if A ne . then do;
    if A &cutoff and AB ne 1 then A1 = 1; else A1 = 0;
  end;

  if B ne . then do;
    if B &cutoff and AB ne 1 then B1 = 1; else B1 = 0;
  end;

%end;

%ELSE %IF &venn_diagram = 3 %THEN %DO;

  if A ne . and B ne . and C ne . then do;
    if A &cutoff and B &cutoff and C &cutoff then ABC = 1;
    else ABC = 0;
  end;

  if A ne . and B ne . then do;
    if A &cutoff and B &cutoff and ABC ne 1 then AB = 1;
    else AB = 0;
  end;

  if A ne . and C ne . then do;
    if A &cutoff and C &cutoff and ABC ne 1 then AC = 1;
    else AC = 0;
  end;

  if B ne . and C ne . then do;
    if B &cutoff and C &cutoff and ABC ne 1 then BC = 1;
    else BC = 0;
  end;

  if A ne . then do;

```

```

        if A &cutoff and AB ne 1 and AC ne 1 and ABC ne 1 then A1 = 1;
        else A1 = 0;
    end;

    if B ne . then do;
        if B &cutoff and AB ne 1 and BC ne 1 and ABC ne 1 then B1 = 1;
        else B1 = 0;
    end;

    if C ne . then do;
        if C &cutoff and AC ne 1 and BC ne 1 and ABC ne 1 then C1 = 1;
        else C1 = 0;
    end;

%END;

%ELSE %IF &venn_diagram=4 %THEN %DO;

    if A ne . and B ne . and C ne . and D ne . then do;
        if A &cutoff and B &cutoff and C &cutoff and D &cutoff then ABCD = 1;
        else ABCD = 0;
    end;

    if A ne . and B ne . and C ne . then do;
        if A &cutoff and B &cutoff and C &cutoff and ABCD ne 1 then ABC = 1;
        else ABC = 0;
    end;

    if A ne . and B ne . and D ne . then do;
        if A &cutoff and B &cutoff and D &cutoff and ABCD ne 1 then ABD = 1;
        else ABD = 0;
    end;

    if A ne . and C ne . and D ne . then do;
        if A &cutoff and C &cutoff and D &cutoff and ABCD ne 1 then ACD = 1;
        else ACD = 0;
    end;

    if B ne . and C ne . and D ne . then do;
        if B &cutoff and C &cutoff and D &cutoff and ABCD ne 1 then BCD = 1;
        else BCD = 0;
    end;

    if A ne . and B ne . then do;
        if A &cutoff and B &cutoff and ABC ne 1 and ABD ne 1 and ABCD ne 1 then AB = 1;
        else AB = 0;
    end;

    if A ne . and C ne . then do;
        if A &cutoff and C &cutoff and ABC ne 1 and ACD ne 1 and ABCD ne 1 then AC = 1;
        else AC = 0;
    end;

    if A ne . and D ne . then do;
        if A &cutoff and D &cutoff and ABD ne 1 and ACD ne 1 and ABCD ne 1 then AD = 1;
        else AD = 0;
    end;

    if B ne . and C ne . then do;
        if B &cutoff and C &cutoff and ABC ne 1 and BCD ne 1 and ABCD ne 1 then BC = 1;
        else BC = 0;
    end;

    if B ne . and D ne . then do;
        if B &cutoff and D &cutoff and ABD ne 1 and BCD ne 1 and ABCD ne 1 then BD = 1;
        else BD = 0;
    end;

    if C ne . and D ne . then do;
        if C &cutoff and D &cutoff and ACD ne 1 and BCD ne 1 and ABCD ne 1 then CD = 1;
        else CD = 0;
    end;

```

```

end;

if A ne . then do;
  if A &cutoff and AB ne 1 and AC ne 1 and AD ne 1 and ABC ne 1 and ABD ne 1 and ACD ne 1
    and ABCD ne 1 then A1 = 1; else A1 = 0;
end;

if B ne . then do;
  if B &cutoff and AB ne 1 and BC ne 1 and BD ne 1 and ABC ne 1 and ABD ne 1 and BCD ne 1
    and ABCD ne 1 then B1 = 1; else B1 = 0;
end;

if C ne . then do;
  if C &cutoff and AC ne 1 and BC ne 1 and CD ne 1 and ABC ne 1 and ACD ne 1 and BCD ne 1
    and ABCD ne 1 then C1 = 1; else C1 = 0;
end;

if D ne . then do;
  if D &cutoff and AD ne 1 and BD ne 1 and CD ne 1 and ABD ne 1 and ACD ne 1 and BCD ne 1
    and ABCD ne 1 then D1 = 1; else D1 = 0;
end;

%END;

run;

/*
COUNTING THE ELEMENTS IN EACH GROUP
After the Macro identifies the elements in each group it uses PROC UNIVARIATE
to sum up the number of elements in each group.
The total number of element within the diagram i.e. the union of Groups A, B, C,
and D, and the total number of elements in the dataset i.e. the universal set are
then calculated. This is used to identify the number of elements that fall outside the union.
*/

proc univariate data = Data_reformatted2 noprint;
var AB A1 B1
%if &venn_diagram > 2 %then %do;
  ABC AC BC C1
%end;

%if &venn_diagram > 3 %then %do;
  ABCD ABD ACD BCD AD BD CD D1
%end;
;

output out = data sum sum = sum AB sum = sum_A1 sum = sum_B1
%if &venn_diagram > 2 %then %do;
  sum = sum_ABC sum = sum_AC sum = sum_BC sum = sum_C1
%end;

%if &venn_diagram > 3 %then %do;
  sum = sum_ABCD sum = sum_ABD sum = sum_ACD
  sum = sum_BCD sum = sum_AD sum = sum_BD
  sum = sum_CD sum = sum_D1
%end;
;
run;

/* Counting the number in the universal set */

proc sql noprint;
  create table id_count as
  select count(id) as count_id
  from data_reformatted;
quit;

/* Counting the number inside the union */

data data_sum2;
set data_sum;

```

```

totalinside = sum(sum_AB, sum_A1, sum_B1
%if &venn_diagram > 2 %then %do;
    ,sum_ABC, sum_AC, sum_BC, sum_C1
%end;

%if &venn_diagram > 3 %then %do;
    ,sum_ABCD, sum_ABD, sum_ACD, sum_BCD, sum_AD,
    sum_BD, sum_CD, sum_D1
%end;
);
run;

/*
COUNTING THE ELEMENTS THAT FALL OUTSIDE OF THE UNION
Using the fetch function the values of the total number of elements within the
union and the universal set are fetched from the appropriate datasets and assigned
to a macro-variable. The total number of elements that fall outside the diagram is
then calculated by using %eval to evaluate the arithmetic expression of the number
of elements in the universal set - the number of elements within the union.
*/

/* Calculating the total number of unique ids - so that I can calculate
the number that falls outside of the groups*/

proc sql noprint;
    select count_id into: TN
    from id_count;
quit;

/* Calculating the total number of values that fall within the groups */

proc sql noprint;
    select totalinside into: TI
    from data_sum2;
quit;

/* Calculating the total numbers that fall outside all of the groups */

%let TO = %eval(&TN - &TI);

/* Assigning the sums to macro variables */

proc sql noprint;
    select sum_A1, sum_B1, sum_AB into :A, :B, :AB
    from data_sum2;
quit;

%if &venn_diagram > 2 %then %do;

    proc sql noprint;
        select sum_C1, sum_AC, sum_BC, sum_ABC into :C, :AC, :BC, :ABC
        from data_sum2;
        quit;

    %end;

%if &venn_diagram > 3 %then %do;

    proc sql noprint;
        select sum_D1, sum_AD, sum_BD, sum_CD, sum_ABD, sum_ACD, sum_BCD, sum_ABCD into :D, :AD,
        :BD, :CD, :ABD, :ACD, :BCD, :ABCD
        from data_sum2;
        quit;

    %end;

/* The rest of the macro needs to be done seperately for 2, 3 and 4
way plots */

data test;
do x = 1 to 100;

```

```

y = x;
output;
end;
run;

/***** 2 WAY VENN DIAGRAMS *****/

%if &venn_diagram=2 %then %do;

  proc template;
    define statgraph Venn2Way;
      begingraph / drawspace=datavalue;

        /* Plot */
        layout overlay / yaxisopts = (display = NONE) xaxisopts = (display = NONE);
          scatterplot x=x y=y / markerattrs=(size = 0);

          /* Venn Diagram (Circles) */
          drawoval x=36 y=50 width=45 height=60 /
            display=all fillattrs=(color=red)
            transparency=0.75 WIDTHUNIT= Percent HEIGHTUNIT= Percent;

          drawoval x=63 y=50 width=45 height=60 /
            display=all fillattrs=(color=green)
            transparency=0.75 WIDTHUNIT= Percent HEIGHTUNIT= Percent;

          /* Numbers */
          drawtext "&A" / x=33 y=50 anchor=center;
          drawtext "&AB" / x=50 y=50 anchor=center;
          drawtext "&B" / x=66 y=50 anchor=center;
          drawtext "Outside Union - &TO" / x=50 y=10 anchor=center width = 30;

          /* Labels */
          drawtext "&GroupA" / x=30 y=15 anchor=center width = 30;
          drawtext "&GroupB" / x=70 y=15 anchor=center width = 30;

        endlayout;
      endgraph;
    end;
  run;

  ods graphics on / reset = all border = off width=16cm height=12cm imagefmt = png imagename =
"&outputfilename";

  ods listing gpath = "&out_location" image_dpi = 200;

  proc sgrender data=test template=Venn2Way;
  run;

  ods listing close;
  ods graphics off;

%end;

/***** 3 WAY VENN DIAGRAMS *****/

%if &venn_diagram = 3 %then %do;

  proc template;
    define statgraph Venn3Way;
      begingraph / drawspace=datavalue;

        /* Plot */
        layout overlay / yaxisopts = (display = NONE) xaxisopts = (display = NONE);
          scatterplot x=x y=y / markerattrs=(size = 0);

          /* Venn Diagram (Circles) */
          drawoval x=37 y=40 width=45 height=60 /
            display=all fillattrs=(color=red)
            transparency=0.75 WIDTHUNIT= Percent HEIGHTUNIT= Percent;
          drawoval x=63 y=40 width=45 height=60 /

```

```

display=all fillattrs=(color=green)
transparency=0.75 WIDTHUNIT= Percent HEIGHTUNIT= Percent;
drawoval x=50 y=65 width=45 height=60 /
display=all fillattrs=(color=blue)
transparency=0.75 WIDTHUNIT= Percent HEIGHTUNIT= Percent;

/* Numbers */
drawtext "&A" / x=32 y=35 anchor=center;
drawtext "&AB" / x=50 y=30 anchor=center;
drawtext "&B" / x=68 y=35 anchor=center;
drawtext "&ABC" / x=50 y=50 anchor=center;
drawtext "&AC" / x=37 y=55 anchor=center;
drawtext "&BC" / x=63 y=55 anchor=center;
drawtext "&C" / x=50 y=75 anchor=center;
drawtext "Outside Union - &TO" / y=3 x=50 anchor=center width = 30;

/* Labels */
drawtext "&GroupA" / x=30 y=7 anchor=center width = 30;
drawtext "&GroupB" / x=70 y=7 anchor=center width = 30;
drawtext "&GroupC" / x=50 y=98 anchor=center width = 30;

        endlayout;
    endgraph;
end;
run;

ods graphics on / reset = all border = off width=16cm height=12cm imagefmt = png imagename =
"&outputfilename";

ods listing gpath = "&out_location" image_dpi = 200;

proc sgrender data=test template=Venn3Way;
run;

ods listing close;
ods graphics off;

%end;

/***** 4 WAY VENN DIAGRAMS *****/

%if &venn_diagram = 4 %then %do;

proc template;
  define statgraph Venn4Way;
    begingraph / drawspace=datavalue;

      /* Plot */
      layout overlay / yaxisopts = (display = NONE) xaxisopts = (display = NONE);
      scatterplot x=x y=y / markerattrs=(size = 0);

      /* Venn Diagram (Ellipses) */
      drawoval x=28 y=39 width=26 height=100 /
      display=all fillattrs=(color=red transparency=0.85)
outlineattrs=(color=red) transparency=0.50
      WIDTHUNIT= Percent HEIGHTUNIT= Percent rotate = 45 ;
      drawoval x=72 y=39 width=26 height=100 /
      display=all fillattrs=(color=green transparency=0.85)
outlineattrs=(color=green) transparency=0.50
      WIDTHUNIT= Percent HEIGHTUNIT= Percent rotate = 315 ;
      drawoval x=57 y=54 width=26 height=100 /
      display=all fillattrs=(color=blue transparency=0.85)
outlineattrs=(color=blue) transparency=0.50
      WIDTHUNIT= Percent HEIGHTUNIT= Percent rotate = 335 ;
      drawoval x=43 y=54 width=26 height=100 /
      display=all fillattrs=(color=yellow transparency=0.85)
outlineattrs=(color=yellow) transparency=0.50
      WIDTHUNIT= Percent HEIGHTUNIT= Percent rotate = 25 ;

      /* Numbers */

```

```

y=60 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&A" / x=13
y=80 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&B" / x=35
y=80 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&C" / x=65
y=60 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&D" / x=87

y=45 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&AB" / x=36
y=16 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&AC" / x=41
y=6 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&AD" / x=50
y=55 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&BC" / x=50
y=16 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&BD" / x=59
y=45 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&CD" / x=64

y=30 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&ABC" / x=43
y=30 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&BCD" / x=57
y=12 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&ACD" / x=46
x=52 y=12 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&BCD" /

x=50 y=21 anchor=center; drawtext textattrs = GRAPHVALUETEXT(size = 6pt weight = bold) "&ABCD" /
Union - &TO" / x=70 y=5 anchor=center width = 30;
/* Labels */

drawtext textattrs = GRAPHVALUETEXT(size = 7pt weight = bold) "&GroupA" / x=6
y=20 anchor=center width = 30;
drawtext textattrs = GRAPHVALUETEXT(size = 7pt weight = bold) "&GroupB" / x=6
y=85 anchor=center width = 30;
drawtext textattrs = GRAPHVALUETEXT(size = 7pt weight = bold) "&GroupC" / x=82
y=85 anchor=center width = 30;
drawtext textattrs = GRAPHVALUETEXT(size = 7pt weight = bold) "&GroupD" / x=82
y=20 anchor=center width = 30;

endlayout;
endgraph;
end;
run;

ods graphics on / reset = all border = off width=16cm height=12cm imagefmt = png imagename =
"&outputfilename";

ods listing gpath = "&out_location" image_dpi = 200;

proc sgrender data=test template=Venn4Way;
run;

ods listing close;
ods graphics off;

%end;

%mend venn;

/* Example macro calls to produce 2,3 and 4 way diagrams */

%venn( data = work.data);

```